

Российский университет дружбы народов имени Патриса Лумумбы

Дисциплина «Имитационное моделирование»

Лабораторная работа № 6

Реализация SIR в подходе сетей Петри

Выполнила: Куокконен Дарина Андреевна

Группа: НФИбд-01-23

Москва, 2026

Содержание

1. Цель и постановка задачи.
2. Модель SIR и сеть Петри.
3. Вычислительная среда и общий модуль.
4. Базовые сценарии: ОДУ, SSA, сканирование, анимация, анализ CSV.
5. Параметрические варианты четырёх сценариев.
6. Литературное программирование и выполненные блокноты.
7. Проверки, ограничения и выводы.

Аннотация. Построена сеть Петри SIR и сопоставлены её непрерывная и событийная интерпретации. Выполнены четыре основных сценария главы 6 практикума и четыре параметрических варианта. Сохранены закон массового действия βSI , исходные параметры и временные сетки задания. Отдельно исследовано влияние шага сохранения на оценку пика заражений. Числа в отчёте получены из сохранённых результатов Julia.

1. Цель и постановка задачи

Цель — освоить представление динамической системы сетью Петри и два способа её исполнения: систему обыкновенных дифференциальных уравнений и стохастический процесс с дискретными событиями. Проверяется согласованность структуры сети, численного решения, событийной траектории и анализа CSV.

Выполняется глава 6 «Реализация основных моделей в подходе сетей Петри» практикума А. В. Корольковой и Д. С. Кулябова, стр. 191–207. Основное задание содержит общий модуль `src/SIRPetri.jl` и четыре самостоятельных сценария. §6.2 требует литературные исходники, производные форматы, параметрические варианты и интеграцию документации в отчёт.

Таблица 1: Основные сценарии задания


Сценарий	Назначение	Результат
<code>sirpetri_run.jl</code>	ОДУ и SSA	Траектории S/I/R, CSV и сеть
<code>sirpetri_scan_parameters.jl</code>	15 значений β	Пик I и R(100)
<code>sirpetri_animate.jl</code>	Изменение маркировок	GIF из 501 кадра
<code>sirpetri_report.jl</code>	Чтение сохранённых CSV	Сравнение I(t) и чувствительности

Четыре файла с суффиксом `__param` расширяют каждый из этих сценариев, не заменяя базовые запуски. Общий модуль, тесты, настройка среды и `tangle.jl` не считаются дополнительными экспериментами.

2. Модель SIR и сеть Петри

2.1 Состояния и переходы

S обозначает восприимчивых, I — заражённых, R — выздоровевших. Рождения, миграция, смертность и повторное заражение не рассматриваются. Позициям сети соответствуют состояния, а переходам — заражение и выздоровление:



В первом переходе S уменьшается на единицу, I увеличивается на единицу; второй переносит одну единицу из I в R . В SSA маркировки целочисленны, в непрерывной интерпретации — действительные неотрицательные величины.




Рисунок 1: Сеть SIR из объекта AlgebraicPetri, построенная Graphviz.

Схема (Рисунок 1) показывает две выходные дуги к I , обеспечивающие суммарное приращение $+1$. Матрица изменения маркировки вычисляется из дуг сети:

$$C = \begin{pmatrix} -1 & 0 \\ 1 & -1 \\ 0 & 1 \end{pmatrix}.$$

Суммы столбцов равны нулю: $N = S + I + R$ сохраняется при каждом переходе. Начальная маркировка $(990, 10, 0)$ соответствует $N = 1000$.

2.2 Непрерывная и событийная интерпретации

Интенсивности массового действия равны $a_1 = \beta SI$ и $a_2 = \gamma I$. В работе **нет деления βSI на N** . Система уравнений имеет вид

$$\dot{S} = -\beta SI, \dot{I} = \beta SI - \gamma I, \dot{R} = \gamma I.$$

Начальный рост I определяется $\beta S_0 > \gamma$. При $\gamma = 0,1$ критическое значение β равно $0,1/990 \approx 0,00010101$. Весь диапазон $\beta = 0,1 \dots 0,8$ находится выше порога. Утверждение примера на стр. 201 об отсутствии эпидемии при $\beta = 0,1$ противоречит указанному закону. Замена на $\beta SI/N$ дала бы другую модель и поэтому не производилась.

Прямой метод Гиллеспи генерирует время до события из экспоненциального распределения с параметром $a_0 = a_1 + a_2$. Переход выбирается с вероятностями a_1/a_0 и a_2/a_0 . После события интенсивности пересчитываются. Событие за пределами T не исполняется; последнее состояние дополняется строкой T . Если $a_0 = 0$, система остаётся в текущей маркировке.

2.3 Независимый контроль пика

Для непрерывной модели выполняется интегральное соотношение

$$I + S - \frac{\gamma}{\beta} \ln S = I_0 + S_0 - \frac{\gamma}{\beta} \ln S_0.$$

Внутренний максимум достигается при $S^* = \gamma/\beta$, откуда

$$I_{max} = I_0 + S_0 - S_p + S_p \ln(S_p/S_0), S_p = \gamma/\beta.$$

Формула проверяет численное решение независимо от интегратора. Она не подменяет требуемый `maximum(df.I)` на сетке 0,5: дискретный и уточнённый пики далее приводятся отдельно.

3. Вычислительная среда и общий модуль

3.1 Организация проекта

Использована Julia 1.11.9. Версии закреплены в `Project.toml` и `Manifest.toml`: AlgebraicPetri 0.8.11, Catlab 0.14.17, OrdinaryDiffEq 7.8.1, DataFrames 1.8.2, CSV 0.10.17, Plots 1.41.7, DrWatson 2.19.1, Literate 2.21.0, IJulia 1.34.4. Общий модуль расположен в `project/src`, литературные исходники — в `project/scripts`, данные — в `project/data`. Все иллюстрации находятся в едином каталоге `image`.

```
julia --version
julia --project=. scripts/setup_environment.jl
make tangle
make run
make notebooks
make docs
make test
```

`make run` исполняет восемь сценариев; `make notebooks` отдельно запускает их блокноты ядром Julia. DrWatson восстанавливает корень проекта независимо от вложенного каталога блокнота.

```
dakuokkonen@lab06:/workspace/labs/lab06/project$ whoami
dakuokkonen
dakuokkonen@lab06:/workspace/labs/lab06/project$ julia --version
julia version 1.11.9
dakuokkonen@lab06:/workspace/labs/lab06/project$ julia --project=. -e 'using Pkg; Pkg.status()'
Project project v6.0.0
Status `~/workspace/labs/lab06/project/Project.toml`
⚠ [4f99eebe] AlgebraicPetri v0.8.11
[336ed68f] CSV v0.10.17
⚠ [134e5e36] Catlab v0.14.17
[a93c6f00] DataFrames v1.8.2
[634d3b9d] DrWatson v2.19.1
[7073ff75] IJulia v1.34.4
[98b081ad] Literate v2.21.0
[1dea7af3] OrdinaryDiffEq v7.8.1
[91a5bcdd] Plots v1.41.7
[10745b16] Statistics v1.11.4
[9a3f8284] Random v1.11.0
[8dfed614] Test v1.11.0
Info Packages marked with ⚠ have new versions available but compatibility constraints restrict them from
upgrading. To see why use `status --outdated`
dakuokkonen@lab06:/workspace/labs/lab06/project$
```

Рисунок 2: Пользователь *dakuokkonen*, версия Julia и установленные пакеты проекта.

Снимок (Рисунок 2) фиксирует фактически используемое окружение.

3.2 Построение сети и правой части

Листинг из `project/src/SIRPetri.jl` показывает настоящий объект `LabelledPetriNet`. Преобразование `PetriNet(net)` снимает символьные метки, сохраняя структуру, и согласует индексацию с числовым вектором состояний.

```
function build_sir_network(β=0.3, γ=0.1)
    all(isfinite, (β, γ)) && min(β, γ) >= 0 ||
        throw(ArgumentError("Invalid rates"))
    net = LabelledPetriNet([:S, :I, :R],
        :infection => (:S, :I) => (:I, :I)),
        :recovery => (:I => :R))
    net, [990.0, 10.0, 0.0], [:S, :I, :R]
end

function sir_ode(net, rates=[0.3, 0.1])
    f! = vectorfield(PetriNet(net))
    (du, u, p, t) -> f!(du, u, rates, t)
end
```

Правая часть получается вызовом `vectorfield`, а не независимой ручной записью уравнений. Tsit5 использует относительный допуск 10^{-9} и абсолютный 10^{-10} ; проверяется успешное завершение решателя. Для уточнения пика допуски ужесточены до 10^{-10} и 10^{-11} .

3.3 Прямой метод Гиллеспи

Цикл из `simulate_stochastic` в `src/SIRPetri.jl` приведён ниже. До него проверяются параметры, начальная целочисленная маркировка и интервал. Матрица `C` извлекается из дуг сети.

```
while t < tend
    propensities = [rates[j] * prod(u[s] for s in inputs(net, j))
                    for j in 1:nt(net)]
    total = sum(propensities)
    total > 0 || break
    next_t = t - log(rand(rng)) / total
    next_t <= tend || break
    transition = searchsortedfirst(cumsum(propensities), rand(rng) *
total)
    u .+= C[:, transition]
    minimum(u) >= 0 || error("Negative token count")
    t = next_t
    push!(df, (t, u[1], u[2], u[3]))
end
df.time[end] < tend && push!(df, (tend, u[1], u[2], u[3]))
```

Во входах переходов SIR нет повторяющихся позиций, поэтому произведение маркировок даёт βSI и γI . Модуль не заявляется универсальным SSA для реакций с произвольными кратностями одинаковых реагентов.

4. Основные вычислительные сценарии

4.1 Базовый опыт

Для `sirpetri_run.jl` взяты $\beta=0,3$, $\gamma=0,1$, $T=100$, шаг сохранения ОДУ 0,5, `seed=123` для SSA. CSV содержат все столбцы `time, S, I, R`, а не одну агрегированную метрику. Траектория ОДУ имеет 201 строку.

Ниже включена документация, автоматически полученная из `project/scripts/sirpetri_run.jl` средствами `Literate`. Аналогичный способ использован для всех восьми сценариев: QMD-фрагменты подключены непосредственно в соответствующие разделы отчёта.

4.1.1 Базовый опыт: ОДУ и прямой метод Гиллеспи

Источник: §6.1.3 практикума, стр. 197–200. Используется βSI , без деления на N .

```

using DrWatson
@quickactivate "project"
include(sourcedir("SIRPetri.jl"))
using .SIRPetri
include(sourcedir("experiment_support.jl"))
using Catlab.Graphics
import Catlab.Graphics.Graphviz

```

4.1.1.1 Сеть и фиксированные параметры

```

β, γ, tmax = 0.3, 0.1, 100.0
net, u0, states = build_sir_network(β, γ)
stoichiometry(net)

```

4.1.1.2 Детерминированная и стохастическая траектории

```

df_det = simulate_deterministic(net, u0, (0.0, tmax); saveat=0.5,
rates=[β, γ])
Random.seed!(123)
df_stoch = simulate_stochastic(net, u0, (0.0, tmax); rates=[β, γ])
save_table("sir_det.csv", df_det)
save_table("sir_stoch.csv", df_stoch)
summary = DataFrame([merge((method="ODE",), epidemic_summary(df_det)),
merge((method="SSA",), epidemic_summary(df_stoch))])
save_table("sir_summary.csv", summary)
summary

```

4.1.1.3 Динамика трёх состояний

```

p_det = plot_sir(df_det; title="Deterministic SIR: beta=0.3,
gamma=0.1")
savefig(p_det, imagepath("02-plot.png"))
display(p_det)
p_stoch = plot_sir(df_stoch; stochastic=true, title="Stochastic SIR:
seed=123")
savefig(p_stoch, imagepath("03-plot.png"))
display(p_stoch)

```

4.1.1.4 Структура сети Петри

```

open(imagepath("01-plot.png"), "w") do io
    Graphviz.run_graphviz(io, to_graphviz_sir(net); format="png")
end

```

4.1.1.5 Ранний пик и ошибка дискретизации наблюдений

Шаг .5 оставлен как в методичке. Уточнение пика является отдельной проверкой.

```

fine = simulate_deterministic(net, u0, (0.0, 0.2); saveat=0.0002,
rates=[β, γ])
peak = refined_peak(net, u0, (0.0, tmax); rates=[β, γ])
save_table("sir_early.csv", fine)

```



```

save_table("sir_peak_refined.csv", DataFrame([peak]))
p_early = plot_sir(fine; title="Early transient: the 0.5 grid misses
the peak")
savefig(p_early, imagepath("04-plot.png"))
display(p_early)
DataFrame([peak])

```




Рисунок 3: Детерминированные S , I и R на заданном горизонте.




Рисунок 4: Событийные S , I и R , $seed=123$.

Графики (Рисунок 3 и Рисунок 4) показывают, что восприимчивые быстро заражаются, после чего I убывает вследствие выздоровления, а R возрастает. SSA изображена ступенчато: отдельное событие меняет маркировку дискретно.

Таблица 2: Базовый опыт, *sir_summary.csv*

Показатель	ОДУ, сетка 0,5	SSA, события
Максимум I	952,691393	999
Время максимума	0,5	0,037042
$R(100)$	999,954530	1000
Ошибка сохранения N	$2,16 \cdot 10^{-12}$	0

В базовом сравнении (Таблица 2) способы наблюдения различаются. Уточнение ОДУ даёт $t=0,0420463125$ и $I=997,0012275894$, совпадая с аналитическим значением с ошибкой менее 10^{-8} . Исходная сетка недооценивает пик примерно на 44,31 человека, или 4,44%; это не ошибка интегратора.




Рисунок 5: Ранний переход на сетке 0,0002.

Рисунок 5 показывает, что $t=0,5$ находится уже после максимума. Дополнительные данные записаны в `sir_early.csv` и `sir_peak_refined.csv`; они не заменяют обязательные `sir_det.csv` и `sir_stoch.csv`.

4.2 Сканирование коэффициента заражения

`sirpetri_scan_parameters.jl` выполняет 15 запусков $\beta=0,1:0,05:0,8$ при $\gamma=0,1$, $T=100$, шаг 0,5 и прежних начальных условиях. `sir_scan.csv` содержит β , `peak_I` и `final_R`.

4.2.1 Чувствительность к коэффициенту заражения

§6.1.4, стр. 200–202: ровно 15 значений β и заданная сетка наблюдений .5.

```
using DrWatson
@quickactivate "project"
include(scrdir("SIRPetri.jl"))
using .SIRPetri
include(scrdir("experiment_support.jl"))
```

4.2.1.1 Сканирование при $\gamma=.1$

```
 $\beta$ _range = 0.1:0.05:0.8
 $\gamma$ _fixed, tmax = 0.1, 100.0
df_scan = DataFrame( $\beta$ =Float64[], peak_I=Float64[], final_R=Float64[])
df_refined = DataFrame( $\beta$ =Float64[], peak_I=Float64[],
```

```

peak_time=Float64[], analytic=Float64[])
for  $\beta$  in  $\beta$ _range
    net, u0, _ = build_sir_network( $\beta$ , y_fixed)
    df = simulate_deterministic(net, u0, (0.0, tmax); saveat=0.5,
rates=[ $\beta$ , y_fixed])
    push!(df_scan, ( $\beta$ , maximum(df.I), df.R[end]))
    peak = refined_peak(net, u0, (0.0, tmax); rates=[ $\beta$ , y_fixed])
    push!(df_refined, ( $\beta$ , peak.peak, peak.time, peak.analytic))
end
save_table("sir_scan.csv", df_scan)
save_table("sir_scan_refined.csv", df_refined)
df_scan

```

4.2.1.2 Обязательные показатели на заданной сетке

```

p = plot(df_scan. $\beta$ , [df_scan.peak_I df_scan.final_R];
    label=["Peak I, saveat=0.5" "Final R"], marker=:circle,
    xlabel="beta", ylabel="Population", title="Sensitivity to
infection rate")
savefig(p, imagepath("05-plot.png"))
display(p)

```

4.2.1.3 Уточнённый пик не равен максимуму дискретных отсчётов

```

p_peak = plot(df_refined. $\beta$ , df_refined.peak_I; marker=:circle,
    label="Refined peak", xlabel="beta", ylabel="Peak I", title="Early
epidemic peak")
savefig(p_peak, imagepath("06-plot.png"))
display(p_peak)
df_refined

```




Рисунок 6: Дискретный пик I и конечное R для обязательного диапазона β .




Рисунок 7: Уточнённая высота пика при изменении β .

При $\beta=0,1$ дискретный максимум равен 955,6260, при 0,3 — 952,6914, при 0,8 — 951,7772. Убывание дискретной оценки (Рисунок 6) не означает ослабления эпидемии: увеличение β сдвигает пик раньше, и к $t=0,5$ выздоровление успевает сильнее уменьшить

I. Уточнённый пик (Рисунок 7), напротив, возрастает; $R(100)$ почти постоянно и близко к 1000.

4.3 Анимация маркировок

`sirpetri_animate.jl` использует $\beta=0,3$, $\gamma=0,1$ и 501 момент от 0 до 100 с шагом 0,2. В кадре три столбца S/I/R с фиксированной шкалой 0–1050. `01-animation.gif` при 25 кадрах/с длится около 20 секунд.

4.3.1 Анимация детерминированной динамики

§6.1.5, стр. 202–204. В печатном коде ошибочно повторён сканирующий сценарий. Здесь реализовано текстовое задание: 501 кадр S/I/R на всём интервале 0...100.

```
using DrWatson
@quickactivate "project"
include(sourcedir("SIRPetri.jl"))
using .SIRPetri
include(sourcedir("experiment_support.jl"))
```

4.3.1.1 Трактория с шагом .2

```
net, u0, _ = build_sir_network(0.3, 0.1)
df_animation = simulate_deterministic(net, u0, (0.0, 100.0);
saveat=0.2, rates=[0.3, 0.1])
save_table("sir_animation.csv", df_animation)
first(df_animation, 6)
```

4.3.1.2 Сохранение и показ GIF

```
animation = animate_sir(df_animation, "01-animation.gif";
title="beta=0.3, gamma=0.1")
display(animation)
```

4.3.1.3 Опорные состояния для печатного документа

```
panels = [bar(["S", "I", "R"], [df_animation.S[i], df_animation.I[i],
df_animation.R[i]]);
color=[:royalblue, :firebrick, :seagreen], legend=false,
ylim=(0,1050),
title="t=$(df_animation.time[i])" for i in [1,2,51,501]]
p_frames = plot(panels...; layout=(2,2), size=(1100,750))
savefig(p_frames, imagepath("07-plot.png"))
display(p_frames)
```




Рисунок 8: Состояния анимации при $t=0; 0,2; 10; 100$.

В последовательности кадров (Рисунок 8) сначала преобладает S, затем I, затем R. Этот рисунок служит печатным представлением реально построенной GIF. Шаг 0,2 также не разрешает ранний пик подробно. На стр. 203–204 примера вместо анимации повторён код сканирования; здесь выполнено текстовое требование об изменении маркировок во времени.

4.4 Отдельный анализ CSV

`sirpetri_report.jl` читает `sir_det.csv`, `sir_stoch.csv` и `sir_scan.csv` без повторной симуляции. SSA использует свои реальные времена событий. Для численной метрики она отображается на 201 момент ОДУ правосторонне непрерывной ступенчатой интерполяцией.

4.4.1 Анализ ранее сохранённых результатов

§6.1.6, стр. 204–207. Моделирование здесь не повторяется.

```
using DrWatson
@quickactivate "project"
include(scrdir("SIRPetri.jl"))
using .SIRPetri
include(scrdir("experiment_support.jl"))
```

4.4.1.1 Независимое чтение трёх обязательных CSV

```
df_det = CSV.read(datadir("sir_det.csv"), DataFrame)
df_stoch = CSV.read(datadir("sir_stoch.csv"), DataFrame)
df_scan = CSV.read(datadir("sir_scan.csv"), DataFrame)
(nrow(df_det), nrow(df_stoch), nrow(df_scan))
```

4.4.1.2 Сравнение $I(t)$ по действительным временам

```
p_comparison = plot(df_det.time, df_det.I; label="ODE", xlabel="Time",
  ylabel="Infected", title="ODE and SSA: original time grids")
plot!(p_comparison, df_stoch.time, df_stoch.I; label="SSA",
  seriestype=:steppost)
savefig(p_comparison, imagepath("08-plot.png"))
display(p_comparison)
```

4.4.1.3 Численная разность на общей сетке

```
aligned = state_at(df_stoch, df_det.time)
difference = DataFrame(time=df_det.time, ODE_I=df_det.I,
  SSA_I=aligned.I,
  difference=aligned.I .- df_det.I)
save_table("sir_comparison.csv", difference)
metrics = DataFrame(rmse_I=[sqrt(mean(difference.difference.^2))],
  max_abs_I=[maximum(abs.(difference.difference))])
save_table("sir_comparison_summary.csv", metrics)
metrics
```

4.4.1.4 Чувствительность без повторного расчёта

```
p_sensitivity = plot(df_scan.β, df_scan.peak_I; marker=:circle,
  label=false,
  xlabel="beta", ylabel="Peak I on 0.5 grid", title="Sensitivity
  from saved CSV")
savefig(p_sensitivity, imagepath("09-plot.png"))
display(p_sensitivity)
```





Рисунок 9: $I(t)$ ОДУ и SSA на собственных временных сетках.




Рисунок 10: Чувствительность дискретного пика, восстановленная из CSV.

При сравнении (Рисунок 9) наблюдается близкая динамика после быстрого начального перехода. RMSE на общей сетке равна 3,5954 человека, максимальное абсолютное различие — 19,0308. Это характеристика одного SSA, не доверительный интервал и не ошибка интегратора. Рисунок 10 воспроизводит метрику сканирования из

сохранённых данных. Ошибка согласования временных индексов в примере стр. 206 исправлена.

5. Параметрические варианты

5.1 Девять сочетаний и 180 стохастических запусков

Для §6.2 приняты $\beta \in \{0,1;0,3;0,8\}$ и $\gamma \in \{0,05;0,1;0,2\}$. Каждому из девяти сочетаний соответствуют ОДУ и 20 SSA. Начальные условия, $T=100$ и шаг ОДУ 0,5 сохранены. Seed равен $123+100*case_id+replicate$. Среднее и выборочное стандартное отклонение вычисляются по 20 повторам.

5.1.1 Параметрическая серия и стохастические повторы

Дополнение §6.2: девять сочетаний β/γ , по 20 независимых SSA-траекторий.

```
using DrWatson
@quickactivate "project"
include(sourcedir("SIRPetri.jl"))
using .SIRPetri
include(sourcedir("experiment_support.jl"))
```

5.1.1.1 План: 3×3 сочетания и воспроизводимые seed

```
betas, gammas = [0.1, 0.3, 0.8], [0.05, 0.1, 0.2]
repeats = 20
rows = NamedTuple[]
det_rows = NamedTuple[]
for (case_id, ( $\beta$ ,  $\gamma$ )) in enumerate(Iterators.product(betas, gammas))
    net, u0, _ = build_sir_network( $\beta$ ,  $\gamma$ )
    det = simulate_deterministic(net, u0, (0.0, 100.0); saveat=0.5,
rates=[ $\beta$ ,  $\gamma$ ])
    peak = refined_peak(net, u0, (0.0, 100.0); rates=[ $\beta$ ,  $\gamma$ ])
    push!(det_rows, merge((case_id=case_id,  $\beta$ = $\beta$ ,  $\gamma$ = $\gamma$ ,
refined_peak=peak.peak), epidemic_summary(det)))
    save_table("parameters/det_$(case_id).csv", det)
    for repetition in 1:repeats
        seed = 123 + 100case_id + repetition
        stoch = simulate_stochastic(net, u0, (0.0, 100.0);
            rates=[ $\beta$ ,  $\gamma$ ], rng=Xoshiro(seed))
        push!(rows, merge((case_id=case_id,  $\beta$ = $\beta$ ,  $\gamma$ = $\gamma$ ,
repetition=repetition, seed=seed),
            epidemic_summary(stoch)))
        repetition == 1 && save_table("parameters/ssa_$(case_id).csv",
stoch)
    end
end
```

```

end
ensemble = DataFrame(rows)
deterministic = DataFrame(det_rows)
save_table("sir_param_replicates.csv", ensemble)
save_table("sir_param_deterministic.csv", deterministic)
first(ensemble, 10)

```

5.1.1.2 Средние и стандартные отклонения, не доверительный интервал

```

summary = combine(groupby(ensemble, [:case_id, :β, :γ]),
  :peak_I => mean => :peak_mean, :peak_I => std => :peak_sd,
  :final_R => mean => :final_R_mean, :final_R => std => :final_R_sd,
  :peak_time => mean => :time_mean)
save_table("sir_param_summary.csv", summary)
summary

```

5.1.1.3 Изменение стохастического пика при разных скоростях

```

p_ensemble = plot(; xlabel="beta", ylabel="Peak infected", title="SSA:
mean and SD, 20 runs")
for γ in gammas
  d = sort(summary[summary.γ .== γ, :], :β)
  plot!(p_ensemble, d.β, d.peak_mean; yerror=d.peak_sd,
marker=:circle, label="gamma=$(γ)")
end
savefig(p_ensemble, imagepath("10-plot.png"))
display(p_ensemble)

```




Рисунок 11: Средние пики SSA и стандартные отклонения по 20 повторам.

Таблица 3: Статистика *sir_param_summary.csv*

β	γ	Средний пик	Станд. отклонение	Среднее R(100)
0,1	0,05	996,85	1,87	993,50
0,3	0,05	998,95	0,76	993,15
0,8	0,05	999,60	0,60	992,80
0,1	0,10	993,35	2,41	999,95
0,3	0,10	997,80	1,36	999,90
0,8	0,10	999,00	1,17	999,95
0,1	0,20	985,70	3,21	1000,00
0,3	0,20	996,00	1,52	1000,00
0,8	0,20	998,55	1,10	1000,00

При фиксированном γ рост β повышает средний пик; рост γ при неизменном β его уменьшает. При $\gamma=0,05$ к $T=100$ остаётся несколько заражённых. $R(100)<1000$ объясняется конечным горизонтом, а не потерей численности. Вертикальные отрезки (Рисунок 11) — стандартные отклонения, не 95%-е интервалы.

5.2 Сетка из 45 опытов

В `sirpetri_scan_parameters__param.jl` исходные 15 β повторены при трёх γ . Максимум на сетке 0,5 и уточнённый пик сохраняются отдельно.

5.2.1 Двухпараметрическое исследование

§6.2: исходные 15 значений β повторяются при $\gamma=.05, .1, .2$; всего 45 ОДУ.

```
using DrWatson
@quickactivate "project"
include(sourcedir("SIRPetri.jl"))
using .SIRPetri
include(sourcedir("experiment_support.jl"))
```

5.2.1.1 Расчёты и отдельное уточнение пика

```
rows = NamedTuple[]
for γ in [0.05, 0.1, 0.2], β in 0.1:0.05:0.8
    net, u0, _ = build_sir_network(β, γ)
    df = simulate_deterministic(net, u0, (0.0, 100.0); saveat=0.5,
rates=[β, γ])
    peak = refined_peak(net, u0, (0.0, 100.0); rates=[β, γ])
    push!(rows, (β=β, γ=γ, peak_I=maximum(df.I), final_R=df.R[end],
        refined_peak=peak.peak, peak_time=peak.time))
end
df_scan_param = DataFrame(rows)
save_table("sir_scan_param.csv", df_scan_param)
first(df_scan_param, 15)
```

5.2.1.2 Две панели: дискретный и уточнённый пик

```
p_grid = plot(; xlabel="beta", ylabel="Peak I, saveat=0.5",
title="Fixed observation grid")
p_refined = plot(; xlabel="beta", ylabel="Refined peak I",
title="Continuous-time peak")
for γ in [0.05, 0.1, 0.2]
    d = df_scan_param[df_scan_param.γ .== γ, :]
    plot!(p_grid, d.β, d.peak_I; marker=:circle, label="gamma=$(γ)")
    plot!(p_refined, d.β, d.refined_peak; marker=:circle,
label="gamma=$(γ)")
end
p = plot(p_grid, p_refined; layout=(1,2), size=(1200,540))
savefig(p, imagepath("11-plot.png"))
display(p)
```




Рисунок 12: Дискретные и уточнённые пики при трёх значениях γ .

Рисунок 12 подтверждает эффект редкой сетки наблюдений для всех трёх γ . Правая панель согласуется с аналитическим условием $S^*=\gamma/\beta$; только по левой панели делать вывод о влиянии β нельзя.

5.3 Три параметрические анимации

Третий вариант использует $\beta=0,1;0,3;0,8$ при $\gamma=0,1$. Каждая GIF содержит 501 кадр на сетке 0,2: всего 1503 кадра в 02-animation.gif, 03-animation.gif и 04-animation.gif. Траектории сохранены в data/parameters, сводка — в sir_animation_param_summary.csv.

5.3.1 Анимации для набора коэффициентов заражения

§6.2: три самостоятельные траектории, каждую показываем на полном горизонте.

```
using DrWatson
@quickactivate "project"
include(sourcedir("SIRPetri.jl"))
using .SIRPetri
include(sourcedir("experiment_support.jl"))
```

5.3.1.1 Три параметра, 501 кадр в каждой GIF

```
animation_rows = NamedTuple[]
for (case_id,  $\beta$ ) in enumerate([0.1, 0.3, 0.8])
    net, u0, _ = build_sir_network( $\beta$ , 0.1)
    df = simulate_deterministic(net, u0, (0.0, 100.0); saveat=0.2,
    rates=[ $\beta$ , 0.1])
    save_table("parameters/animation_$(case_id).csv", df)
    filename = "0$(case_id+1)-animation.gif"
```

```

    anim = animate_sir(df, filename; title="beta=$(β), gamma=0.1")
    display(anim)
    push!(animation_rows, (β=β, γ=0.1, frames=nrow(df), dt=0.2,
end_time=df.time[end]))
end
animation_manifest = DataFrame(animation_rows)
save_table("sir_animation_param_summary.csv", animation_manifest)
animation_manifest

```

Различия сосредоточены в начале процесса; затем почти вся популяция переходит в R. Глобальная шкала полезна для наблюдения выздоровления, но ранний максимум дополнительно следует изучать по численным данным.

5.4 Анализ параметрических CSV

Последний вариант читает пять созданных таблиц, объединяет статистику SSA и уточнённые пики ОДУ по case_id, строит средний пик SSA против ОДУ и зависимость R(100) от β при трёх γ . Симулятор здесь не вызывается.

5.4.1 Анализ сохранённых параметрических опытов

Здесь нет вызовов симулятора: все показатели считываются из ранее созданных CSV.

```

using DrWatson
@quickactivate "project"
include(sourcedir("experiment_support.jl"))

```

5.4.1.1 Источники результатов

```

replicates = CSV.read(datadir("sir_param_replicates.csv"), DataFrame)
summary = CSV.read(datadir("sir_param_summary.csv"), DataFrame)
deterministic = CSV.read(datadir("sir_param_deterministic.csv"),
DataFrame)
scan = CSV.read(datadir("sir_scan_param.csv"), DataFrame)
animation_summary =
CSV.read(datadir("sir_animation_param_summary.csv"), DataFrame)
(replicates=nrow(replicates), combinations=nrow(summary),
scan_points=nrow(scan),
animations=nrow(animation_summary))

```

5.4.1.2 Сопоставление средних SSA и непрерывного пика ОДУ

```

comparison = innerjoin(summary, deterministic[:, [:case_id,
:refined_peak]]; on=:case_id)
comparison.difference = comparison.peak_mean .-
comparison.refined_peak

```

```
save_table("sir_param_comparison.csv", comparison)
comparison
```

5.4.1.3 Пик и конечное число выздоровевших

```
p_peak = plot(comparison.refined_peak, comparison.peak_mean;
  yerror=comparison.peak_sd, seriestype=:scatter, label="Mean SSA
+/- SD",
  xlabel="Refined ODE peak", ylabel="SSA peak", title="Nine
parameter combinations")
lims = extrema(vcat(comparison.refined_peak, comparison.peak_mean))
plot!(p_peak, collect(lims), collect(lims); label="Equality",
linestyle=:dash)
p_final = plot(; xlabel="beta", ylabel="R(100)", title="Finite-horizon
recovered population")
for y in sort(unique(scan.y))
  d = sort(scan[scan.y .== y, :], :β)
  plot!(p_final, d.β, d.final_R; marker=:circle, label="gamma=$(y)")
end
p = plot(p_peak, p_final; layout=(1,2), size=(1200,550))
savefig(p, imagepath("12-plot.png"))
display(p)
```

5.4.1.4 Показатель воспроизводимости данных

```
DataFrame(total_SSA_runs=[nrow(replicates)],
seeds_unique=[length(unique(replicates.seed))],
max_conservation_error=[maximum(replicates.conservation_error)],
total_animation_frames=[sum(animation_summary.frames)])
```




Рисунок 13: Стохастические и непрерывные пики; конечные состояния параметрической серии.

В параметрическом сравнении (Рисунок 13) средние SSA близки к линии равенства, но не обязаны совпадать с пиком ОДУ: максимум — нелинейная статистика, а выборка ограничена 20 повторами. Правая панель показывает влияние γ на $R(100)$.

6. Литературное программирование и блокноты

6.1 Производные форматы

Восемь литературных `.jl` содержат пояснения `Literate` и исполняемый код. `scripts/tangle.jl` формирует чистый Julia-файл, QMD и IPYNB. Фрагмент генератора ниже взят из действительного исходника.

```
Literate.script(source, scriptsdir(name); name, credit=false)
Literate.notebook(source, projectdir("notebooks", name); name,
    execute=false, credit=false)
Literate.markdown(source, dest; name,
    flavor=Literate.QuartoFlavor(), credit=false)
```

Дополнительный `-report.qmd` имеет обычные Julia-блоки и более глубокие заголовки. Его можно включать в отчёт без повторных вычислений при сборке. Все восемь таких включений размещены в соответствующих разделах выше. Листинги общего модуля и тестов дополняют эти включения, а не заменяют их. Документация HTML собирается `make docs`.

6.2 Jupyter

```
jupyter lab
```

В интерфейсе открывается IPYNB из `project/notebooks` с ядром Julia lab06. Последовательно выполняются ячейки настройки, расчёта, таблиц и графиков. Все восемь сохранённых блокнотов содержат счётчики выполнения и выводы без ошибок; анимационные блокноты содержат GIF.




Рисунок 14: Ядро Julia, исходные параметры и матрица изменений сети в выполненном блокноте.




Рисунок 15: Сводная таблица базового опыта и построение графиков в Jupyter.

Снимки [Рисунок 14](#) и [Рисунок 15](#) показывают код вместе со счётчиками выполнения и фактическим выводом ячеек.


7. Проверки и ограничения

7.1 Структурные и численные тесты

В `test/runtests.jl` пройдены 42 проверки: структура и интенсивности — 7, ОДУ — 11, SSA — 14, сохранённые результаты — 10. Проверены сохранение N , неотрицательность с численным допуском для ОДУ, аналитический пик, целочисленность SSA, допустимые события, `seed` и конечный горизонт.

```
@test stoichiometry(net) == [-1 0; 1 -1; 0 1]
@test sum(stoichiometry(net); dims=1) == zeros(1, 2)
@test maximum(abs.(df.S .+ df.I .+ df.R .- 1000)) < 1e-6
@test isapprox(peak.peak, peak.analytic; atol=1e-5)
@test all(a.S .+ a.I .+ a.R .== 1000)
@test all(diff(a.time) .> 0)
@test a.time[end] == 100
@test nrow(runs)==180
@test length(unique(runs.seed))==180
```

Это фрагменты реальных проверок из `project/test/runtests.jl`; инициализация переменных находится в полном файле. Тесты дополняют содержательную сверку задания, но сами по себе не доказывают его полноту.



```
case_id,β,γ,peak_mean,peak_sd,final_R_mean,final_R_sd,time_mean,refined_peak,difference
1,0.1,0.05,996.85,1.8715318802914818,993.5,2.3951705795753164,0.1202835438053947,995.704573938141,1.14542
60618589842
2,0.3,0.05,998.95,0.7591546545162482,993.15,3.166851333488999,0.03995000962768078,998.3850892646042,0.564
9107353958698
3,0.8,0.05,999.6,0.5982430416161186,992.8,2.546411303445653,0.014550372044046056,999.3331066459132,0.2668
9335408684656
4,0.1,0.1,993.35,2.412140346794021,999.95,0.223606797749979,0.10867074176789952,992.1022950568479,1.24770
49431521436
5,0.3,0.1,997.8,1.361114094757441,999.9,0.3077935056255462,0.0412232672605958,997.0012275893922,0.798772
4106077394
6,0.8,0.1,999.0,1.1697953037312037,999.95,0.223606797749979,0.015521414092101024,998.7528566893967,0.2471
4331060329187
7,0.1,0.2,985.7,3.2134585401827866,1000.0,0.0,0.10466203280814672,985.5908844748411,0.10911552515892708
8,0.3,0.2,996.0,1.5217718205053643,1000.0,0.0,0.03909377199358039,994.4645532991635,1.5354467008364736
9,0.8,0.2,998.55,1.0990426455975695,1000.0,0.0,0.014482609940338717,997.6790001739326,0.8709998260673046
dakuokkonen@lab06:/workspace/labs/lab06/project$ less -N test/runtests.jl
dakuokkonen@lab06:/workspace/labs/lab06/project$ make test
julia --project=. test/runtests.jl
Test Summary: | Pass Total Time
Petri structure and rates | 7 7 4.8s
Test Summary: | Pass Total Time
Deterministic solution | 11 11 2.5s
Test Summary: | Pass Total Time
Exact event simulation | 14 14 1.5s
Test Summary: | Pass Total Time
Saved scenario outputs | 10 10 8.1s
dakuokkonen@lab06:/workspace/labs/lab06/project$
```

Рисунок 16: Результаты запуска набора структурных и численных тестов.

Штатный вывод `make test` (Рисунок 16) подтверждает прохождение 42 проверок.

7.2 Ограничения

Модель предполагает однородное перемешивание и постоянные коэффициенты. Время условно; параметры не оценивались по медицинским данным. Сильная начальная вспышка обусловлена ненормированным βSI и не является прогнозом. Шаг сохранения отличается от адаптивного шага Tsit5 и может скрывать экстремумы даже при точном решении. Двадцать повторов не дают оснований утверждать универсальное совпадение SSA и ОДУ или отсутствие редких исходов.

8. Выводы

Построена сеть AlgebraicPetri с тремя позициями и двумя переходами. Из одной структуры получены ОДУ, стехиометрические изменения и интенсивности SSA. Оба исполнения сохраняют $N=1000$, событийное — целочисленные маркировки.

Выполнены четыре основных сценария и четыре параметрических варианта: траектории, сканирование, анимация и отдельный анализ CSV. Расширение включает 180 SSA, 45 точек двухпараметрической сетки и три дополнительные анимации. Производные JL/QMD/IPYNB созданы для каждого сценария; документация интегрирована в тематические разделы отчёта.

Главный результат — необходимость отличать динамику модели от сетки наблюдений. Пик ОДУ около 997 достигается до $t=0,05$ и пропускается шагом 0,5. Противоречия примера исправлены явно: закон βSI сохранён, времена SSA согласованы, а анимация реализует текстовое требование практикума.

9. Источники

1. Королькова А. В., Кулябов Д. С. Имитационное моделирование. Практикум. Глава 6 «Реализация основных моделей в подходе сетей Петри», стр. 191–207. Редакция PDF от 22.05.2026.
2. Исходный код AlgebraicPetri 0.8.11 и Catlab 0.14.17 в закреплённом окружении: LabelledPetriNet, vectorfield, Graphviz.Graph.
3. Исходный код и результаты работы: project/src, project/scripts, project/data, project/notebooks, project/test.
4. project/Manifest.toml: точные версии вычислительной среды.